
	Reason for change:
	In the type definition of IMS_IPCAN_Command, IMS_TestConfiguration is optional:

type record IMS_IPCAN_Command_Type { /* Messages IMS_PTC -> IPCAN
@status APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NBIOT, NR5GC, NR5GC_IRAT, POS, UTRAN) */
   IMS_IPCAN_CommandName_Type   Name,
   IMS_TestConfiguration_Type   TestConfiguration  optional,
   IMS_TestProcedure_Type       TestProcedure  optional
};
Which means that a non-initialized value may be assigned. Given that  v_TestConfiguration  is only used during initialization of the test case when  v_Command.Name  is  IPCAN_INIT , the variable assignment should be limited to this initialization case. Otherwise  v_TestConfiguration  will remain uninitialized and thus provoke a runtime error.

	
	

	Summary of change:
	Assign variable v_TestConfiguration only in case that  
v_Command.Name := IPCAN_INIT.

	
	

	Consequences if not approved:
	May cause runtime errors in non-initialization related cases.

1 Corrections required

1.1 Correction to usage of variable v_TestConfiguration
	Function name
	f_WLAN_IPCAN_MainLoop (),
(+ similar for 
   f_EUTRA_IPCAN_MainLoop ()
   f_NR5GC_IPCAN_MainLoop ()
   f_UTRAN_IPCAN_MainLoop ()

	Reason for change
	In the type definition of IMS_IPCAN_Command, IMS_TestConfiguration is optional:

type record IMS_IPCAN_Command_Type { /* Messages IMS_PTC -> IPCAN
@status APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NBIOT, NR5GC, NR5GC_IRAT, POS, UTRAN) */
   IMS_IPCAN_CommandName_Type   Name,
   IMS_TestConfiguration_Type   TestConfiguration  optional,
   IMS_TestProcedure_Type       TestProcedure  optional
};
Which means that a non-initialized value may be assigned. Given that  v_TestConfiguration  is only used during initialization of the test case when  v_Command.Name  is  IPCAN_INIT , the variable assignment should be limited to this initialization case. Otherwise  v_TestConfiguration  will remain uninitialized and thus provoke a runtime error.

	Summary of change
	Assign variable v_TestConfiguration only in case that  
v_Command.Name := IPCAN_INIT.

	TTCN module
	IMS/Part3_IPCAN/IPCAN_WLAN.ttcn, 


	MCC160 Comment
	


Before Change:

	<<SKIPPED CODE>>
  function f_WLAN_IPCAN_MainLoop() runs on WLAN_PTC
  {
    var WLAN_CellId_Type v_CellId := wlan_Cell27;    /* arbitrarily selected */
    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_Coordination_MSG;
    var IMS_IPCAN_Command_Type v_Command;
    var integer v_PdnIndex;
    var boolean v_LoopbackEstablished := false;
    var boolean v_Done := false;
    var IMS_TestConfiguration_Type v_TestConfiguration;
    while (not v_Done) {
      alt {
        [] IMS[tsc_Index_IMS1].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {
          v_PdnIndex := tsc_Index_IMS1;
        }
        [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {
          v_PdnIndex := tsc_Index_IMS2;
        }
      }
      v_Command := v_IMS_IPCAN_Coordination_MSG.IMS_IPCAN_Command;

      v_TestConfiguration := v_Command.TestConfiguration;

      select (v_Command.Name) {
        case (IPCAN_INIT) {
            select(v_TestConfiguration){
                case (IPCAN_XCAP){
                     if (pc_WLAN_XCAP_without_PDN){ // @sic R5-174498 sic@
                        f_WLAN_IPCAN_Init(IMS[v_PdnIndex], v_CellId,{ePDG_IMS1});
                    }else{ /// @sic R5-174498 36.508 cl.4.5A.25 sic@
                        f_WLAN_IPCAN_Init(IMS[v_PdnIndex], v_CellId,{ePDG_IMS1, ePDG_XCAP});
                    }
                }
                case (IPCAN_EmergencyCall){
                    f_WLAN_IPCAN_Init(IMS[v_PdnIndex], v_CellId,{ePDG_IMS1, ePDG_IMS2}); //@sic R5-183055 sic@
                }
                case else{
                    f_WLAN_IPCAN_Init(IMS[v_PdnIndex], v_CellId, {ePDG_IMS1});
                }
            }
        }
        case (IPCAN_STARTPROCEDURE) {
          if (f_WLAN_IPCAN_StartProcedure(IMS[v_PdnIndex], v_CellId, v_Command.TestProcedure) == NORMAL) {
            v_LoopbackEstablished := true;   /* NOTE: variable is set to true even when there is no speech or video */
          }
        }
        case (IPCAN_ENDPROCEDURE) {
          f_WLAN_IPCAN_EndProcedure(IMS[v_PdnIndex], IP, v_Command.TestProcedure, v_LoopbackEstablished);
          v_LoopbackEstablished := false;
        }
        case (IPCAN_RELEASE) {
          f_WLAN_IPCAN_Release(IMS[v_PdnIndex], v_CellId, ePDG_IMS1);
          v_Done := true;
        }
        case (IPCAN_QUERY) {  /* @sic R5s130133 additional changes: handling of queries from the IMS PTC sic@ */
          IMS[v_PdnIndex].send(cms_IPCAN_IMS_QueryResponse(f_WLAN_IMS_IpcanInfo()));
        }
        case (IPCAN_PROVIDELOCATIONINFORMATION) {
          FatalError(__FILE__, __LINE__, "'IPCAN_PROVIDELOCATIONINFORMATION' not implemented for WLAN");  /* only used in (EUTRA) emergency call test cases */
        }
        case else {
          FatalError(__FILE__, __LINE__, "invalid command");
        }
      }
    }
  }

<<SKIPPED CODE>>

	


After Change:

	<<SKIPPED CODE>>

  function f_WLAN_IPCAN_MainLoop() runs on WLAN_PTC
  {
    var WLAN_CellId_Type v_CellId := wlan_Cell27;    /* arbitrarily selected */
    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_Coordination_MSG;
    var IMS_IPCAN_Command_Type v_Command;
    var integer v_PdnIndex;
    var boolean v_LoopbackEstablished := false;
    var boolean v_Done := false;
    var IMS_TestConfiguration_Type v_TestConfiguration;
    while (not v_Done) {
      alt {
        [] IMS[tsc_Index_IMS1].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {
          v_PdnIndex := tsc_Index_IMS1;
        }
        [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {
          v_PdnIndex := tsc_Index_IMS2;
        }
      }
      v_Command := v_IMS_IPCAN_Coordination_MSG.IMS_IPCAN_Command;

      select (v_Command.Name) {
        case (IPCAN_INIT) {
        v_TestConfiguration := v_Command.TestConfiguration;
            select(v_TestConfiguration){
                case (IPCAN_XCAP){
                     if (pc_WLAN_XCAP_without_PDN){ // @sic R5-174498 sic@
                        f_WLAN_IPCAN_Init(IMS[v_PdnIndex], v_CellId,{ePDG_IMS1});
                    }else{ /// @sic R5-174498 36.508 cl.4.5A.25 sic@
                        f_WLAN_IPCAN_Init(IMS[v_PdnIndex], v_CellId,{ePDG_IMS1, ePDG_XCAP});
                    }
                }
                case (IPCAN_EmergencyCall){
                    f_WLAN_IPCAN_Init(IMS[v_PdnIndex], v_CellId,{ePDG_IMS1, ePDG_IMS2}); //@sic R5-183055 sic@
                }
                case else{
                    f_WLAN_IPCAN_Init(IMS[v_PdnIndex], v_CellId, {ePDG_IMS1});
                }
            }
        }
        case (IPCAN_STARTPROCEDURE) {
          if (f_WLAN_IPCAN_StartProcedure(IMS[v_PdnIndex], v_CellId, v_Command.TestProcedure) == NORMAL) {
            v_LoopbackEstablished := true;   /* NOTE: variable is set to true even when there is no speech or video */
          }
        }
        case (IPCAN_ENDPROCEDURE) {
          f_WLAN_IPCAN_EndProcedure(IMS[v_PdnIndex], IP, v_Command.TestProcedure, v_LoopbackEstablished);
          v_LoopbackEstablished := false;
        }
        case (IPCAN_RELEASE) {
          f_WLAN_IPCAN_Release(IMS[v_PdnIndex], v_CellId, ePDG_IMS1);
          v_Done := true;
        }
        case (IPCAN_QUERY) {  /* @sic R5s130133 additional changes: handling of queries from the IMS PTC sic@ */
          IMS[v_PdnIndex].send(cms_IPCAN_IMS_QueryResponse(f_WLAN_IMS_IpcanInfo()));
        }
        case (IPCAN_PROVIDELOCATIONINFORMATION) {
          FatalError(__FILE__, __LINE__, "'IPCAN_PROVIDELOCATIONINFORMATION' not implemented for WLAN");  /* only used in (EUTRA) emergency call test cases */
        }
        case else {
          FatalError(__FILE__, __LINE__, "invalid command");
        }
      }
    }
  }

<<SKIPPED CODE>>


